Complex Calculations during Cardiopulmonary Bypass—1987 Technology

Erik G. Deurell, Susan M. Haubert, Alan M. Becker, Jill L. Salamy, Alfred C. Nicolosi, Eric A. Rose, and Henry M. Spotnitz

Columbia University
College of Physicians and Surgeons
New York, NY

Keywords: technique, computer-assisted, anticoagulation, management, computer

Abstract

***(J. Extra-Corpor. Technol.* 19[4]: p. 408-411 Winter 1987*).* Programmable calculators have become increasingly valuable to the conduct of cardiopulmonary bypass (CPB). We report the upgrading of our previous system with current technology. A Sharp PC-2500 portable microcomputer was chosen because of its low cost, easy programmability, rechargeable battery, and built-in printer. This system has allowed perfusionists to perform an increasing number of complex calculations with more accuracy and efficiency. In addition, it offers expanded memory and a permanent record of its calculations. We have used the new system in over 250 cases of CPB without problem. Although the flexibility of such a system allows customization to fit individual needs, we advise that any program be carefully tested and debugged to prevent serious mishap.**

Introduction

In a previous communication[1] we discussed the use of a programmable calculator by perfusionists for certain complex calculations during cardiopulmonary bypass. These calculations included heparin-protamine doses, body surface area, flow for cardiopulmonary bypass, and infusion rates for vasoactive drugs. Although the techniques described continue to be useful, the programmable calculator we employed in 1979 is now outmoded by the availability of inexpensive microcomputers. Therefore, it appeared desirable to update our equipment and improve our software. In selecting a programmable computer, we sought the following characteristics:

1. Easy programmability in a common computer language (eg. BASIC)
2. Power supplied by a rechargeable battery
3. Built-in printer
4. Ability to store and retrieve programs easily and permanently
5. Cost less than $400.00
6. Lightweight, small size—true portability

Having selected what appeared to be an appropriate computer, we then redesigned our software for more generalized utility during cardiopulmonary bypass. In so doing, we attempted to expand the capability of our programs, yet maintain an overall ease of operation that would allow a high degree of interaction with busy perfusionists. Accuracy, immediate utility of calculations, and a hard copy of our records (for review in the event of a questionable result) were also desired.

Materials and Methods

We selected the Sharp portable computer, model PC-2500[a] (Figure 1). The PC-2500 is equipped with both a built-in printer capable of printing graphics and a rechargeable battery. Programs are permanently retained within the computer memory even when the

Supported in Part by USPHS Grant #HL-22894

Direct communications to: Henry M. Spotnitz, M.D., Department of Surgery, Columbia University, College of Physicians and Surgeons, 630 West 168th St., New York, NY 10032.

a Sharp Electronics Corporation, Paramus, NJ 07652



Figure 1: The Sharp portable computer. Model PC-2500.

computer is turned off, as long as the batteries do not completely discharge. In addition, a more permanent recording of programs can be created using a separate audio cassette recorder that allows easy transfer of programs between computers and easy repair if a program is inadvertently damaged. Discount retail cost was less than $300 in late 1986; additional costs include the cassette recorder and the separate 16K removable RAM cartridges that expand the memory of the computer. The computer is programmable in BASIC, making our programs easily adaptable to the requirements of other computer users both within and outside our institution.

Program Operation

The program (Figure 2) prompts the perfusionist to enter the patients' age group, height in inches, weight in pounds, and preoperative hematocrit (HCT). Priming volume (PV) for the pump, intravenous fluids (IF), activated clotting time (ACT) at various intervals, and all heparin or protamine doses are also entered. The age group is then used in conjunction with the height which is converted to centimeters (HT), and weight, which is converted to kilograms (WT), in order to calculate the estimated pump flow (QS) by the formula(e)[1]:

$$\text{0-6 weeks: } QS = 200/WT$$
$$\text{6 weeks-6 months; } QS = 210/WT$$
$$\text{6 months +: } QS = 2.3 * BSA$$

Total body surface area (BSA) is calculated using the formula[2]:

$$BSA = (WT\hat{}.425)*(HT\hat{}.725)*71.84/10000$$

Drug dosages are determined using our modification of the Bull protocol.[3,4] This calculation utilizes a linear regression formula and the patients' ACT at any particular moment to calculate the amount of heparin needed to increase the ACT to 480 seconds, or, in the case of protamine, to return the ACT to its initial (control) value. The linear regression is derived from the rela-

BASIC Computer Program for Cardiopulmonary Bypass Calculations

```
10 DIM ACT(5),HEP(5),C(5): Q=0
20 DIM MHEP(1),MACT(1): DIM DY(3),DX(3)
30 DIM XX(3),YY(3),XY(3)
40 DIM HT(1): DIM WT(1): DIM BSA(1): DIM
   QS(1): CLS: WAIT 0
50 CURSOR 0,1: PRINT "1. 0-6 weeks";
60 CURSOR 0,2: PRINT "2. 6 weeks-6 months";
70 CURSOR 0,3: PRINT "3. 6 months +": WAIT
80 INPUT "Age Group (1,2 or 3): ";AG
90 INPUT "Enter WEIGHT in LBS: ";W1
100 WT=W1/2.2
110 INPUT "Enter HEIGHT in INCHES: ";H1
120 HT=H1*2.54: USING "##.##"
130 BSA=(WT^.425)*(HT^.725)*71.84/10000
140 INPUT "Would you like the post hematocrit
   (Y/N) ";B$: IF B$="N" THEN GOTO 270
150 INPUT "Enter Pre-hematocrit(%): ";HCT:
   LPRINT USING "###.##";"Pre-hematocrit:";HCT;"%"
160 INPUT "Enter Priming volume(ml): ";PV
170 INPUT "Were Whole or Packed cells added(Y/N): ";C$
180 IF C$="Y" THEN INPUT "Whole(W) or Packed(P): ";K$
190 IF C$="N" THEN GOTO 210
200 INPUT "How many cell units were added ";U
210 INPUT "Enter Intravenous fluid(ml): ";IF
   IF BSA<1.09 THEN LET CBV=(80*WT)
220 IF BSA>=1.1 THEN LET CBV=(70*WT)
230 LET HCT=HCT/100:TCV=CBV+PV+IF: RBC=HCT*CBV
240 IF K$="P" THEN RBC=(HCT*CBV)+(175*U)
250 IF K$="W" THEN RBC=(HCT*CBV)+(190*U)
260 LPRINT USING "###.##";"Est. Post-hematocrit "
   ;((RBC/TCV)*100);"%"
270 LPRINT USING "##.##";"BSA=";BSA;" SQ
   METERS":WAIT 59:U=0: K$=""
280 IF AG=1 THEN LET QS=200/WT
290 IF AG=2 THEN LET QS=210/WT
300 IF AG=3 THEN LET QS=2.3*BSA
310 PRINT USING "Pred. Flow=";QS;" L/Min": WAIT 59:
   IF AG=1 THEN LPRINT "Age: 0-6 weeks"
320 IF AG=2 THEN LPRINT "Age: 6 weeks-6 months":
   IF AG=3 THEN LPRINT "Age: 6 months +"
330 LPRINT USING "####";"W=";W1;" HT=";H1;
   USING "####.##";" BSA=";BSA;" QS=";QS
340 INPUT"Enter BASELINE ACT in seconds:";
   ACT(1): HEP(1)=0
350 LPRINT USING "####";"Control ACT=
   ";ACT(1);" sec"
360 LPRINT USING "####";"Suggest 1st HEP
   dose";3*WT;" mg (3 mg/kg)"
370 PRINT "Suggested 1st HEP in mg.:
   ";3*WT;"(3 mg/kg)": WAIT 59
380 INPUT "Enter actual 1st HEP: ";HEP(2)
390 INPUT "Enter ACT in sec 5 min. after 1st
   HEP: ";ACT(2)
400 LPRINT " Actual 1st HEP : ";HEP(2);" mg"
410 C(1)=480-ACT(2): C(2)=ACT(2)-ACT(1)
420 C(3)=(C(1)/C(2))*HEP(2)
430 LPRINT "Est. HEP for ACT=480 sec=";C(3);"
   mg"
440 PRINT "Est. 2nd HEP dose for ACT=480
   sec=";C(3);" mg": WAIT 59
450 INPUT "For LIN. REGRESSION Enter 2nd HEP
   dose (mg) ";HEP(3)
460 INPUT "Enter ACT in sec 5 min. after 2nd
   HEP: ";ACT(3): HEP(3)=HEP(3)+HEP(2)
470 LPRINT "Actual 2nd HEP=";HEP(3)-HEP(2);"
   mg"
480 MACT=(ACT(1)+ACT(2)+ACT(3))/3
490 MHEP=(HEP(2)+HEP(3))/3
500 Z1=0:Z2=0:Z3=0
510 FOR I=1 TO 3
520 DY(I)=ACT(I)-MACT: DX(I)=HEP(I)-MHEP
530 XY(I)=DY(I)*DX(I): XX(I)=DX(I)^2
540 YY(I)=DY(I)^2: Z2=Z2+XX(I)
550 Z3=Z3+YY(I):Z1=Z1+XY(I)
560 NEXT I
570 SL=Z1/Z2
580 PRINT USING "###.##";"Slope=";SL: WAIT
   59: R=Z1/(Z2*Z3)^.5
590 PRINT USING "##.###";"Corr Coeff=";R:
   WAIT 59
600 LPRINT "ADDITIONAL DOSES:": TT=HEP(3)
610 INPUT "Enter HEP for heparin, PROT for
   protamine or 'Q' to quit ";A$
620 IF A$="HEP" THEN GOTO 660
630 IF A$="PROT" THEN GOTO 730
640 IF A$="Q" THEN GOTO 800
650 GOTO 610
660 INPUT "For heparin dose, enter current
   ACT in sec ";T1: S1=(480-T1)/SL
670 LPRINT USING "####";"(Current ACT is
   ";T1;"sec)"
680 PRINT USING "####";"Heparin dose is
   ";S1;" mg": WAIT 59
690 LPRINT "Heparin dose is ";S1;" mg"
700 INPUT "Enter the actual HEP dose given:
   ";H: TT=TT+H
710 LPRINT "Actual Heparin dose given: ";H;"
   mg"
720 LPRINT USING "#####";"Total amount of
   heparin: ";TT;" mg": GOTO 610
730 INPUT "For Protamine, enter current ACT
   in SEC ";T2
740 LPRINT USING "####";"(Current ACT is
   ";T2;"sec)"
750 PRINT USING "####";"Protamine dose is
   ";(T2-ACT(1))*1.3/SL;" mg": WAIT 59
760 LPRINT "Protamine dose is ";(T2-
   ACT(1))*1.3/SL;" mg"
770 INPUT "Actual Protamine dose: ";P
780 LPRINT "Actual Protamine dose given:
   ";P;" mg": Q=Q+P
790 LPRINT USING "#####";"Total amount of
   Protamine given: ";Q;" mg": GOTO 610
800 INPUT "Final ACT: ";A
810 LPRINT "Final ACT: ";A;" sec"
820 LPRINT "Total Heparin given: ";TT;" mg"
830 LPRINT "Total Protamine given: ";Q;" mg"
840 END
```

Figure 2: Listing of the program used in Cardiopulmonary Bypass.

tion of ACT to heparin administered in three steps during initial heparinization for cardiopulmonary bypass. The microcomputer saves time and minimizes error.

An optional feature of the computer program determines the predicted hematocrit on cardiopulmonary bypass. The estimated HCT is the quotient of the red blood cell volume (RBC) and the combined volumes of the circulating blood (CBV), PV, and IF:

$$\text{Est. HCT} = \frac{\text{RBC}}{\text{CBV} + \text{PV} + \text{IF}}$$

RBC is calculated as the product of the preoperative HCT and the CBV. The CBV is derived from the formula(e):

$$\text{BSA} < 1.09;\ \text{CBV} = \text{WT*80}$$
$$\text{BSA} > = 1.1;\ \text{CBV} = \text{WT*70}$$

In addition, if the perfusionist elects to add blood to the priming volume, the program requests the number of units (U) of red blood cells given to the patient and makes the distinction as to whether the units are whole or packed. The RBC is then calculated with the formula(e):

$$\text{whole; RBC} = (\text{HCT*CBV}) + (175\text{*U})$$
$$\text{packed; RBC} = (\text{HCT*CBV}) + (190\text{*U})$$

All of the input and calculated data are printed out onto a data slip, utilizing the built-in printer.

Results

The sample data slip (Figure 3) illustrates the patient's preoperative hematocrit, estimated hematocrit on cardiopulmonary bypass, expected pump flow (liters/min.), body surface area, height (in inches) and weight (in pounds). Dosages of heparin and protamine are also displayed when inputted. Additional doses of heparin and protamine administered are shown with their corresponding ACT and the running tally of the total dosages given. While this information is being placed on a permanent record using a replacable pen, it is also being displayed on the computer's screen. When the program has finished, a table listing the final ACT, total heparin given, and total protamine given is printed.

Although the data slip contains the crux of the information computed by the program, the screen is the actual link between operator and computer. Whether a particular statement is printed depends upon the BASIC command used. Thus, to fully visualize the requirements needed to operate the program, one must look first at the program listing (Figure 2) and the commands utilized. In so doing, one realizes that all of the questions regarding data input are shown on the screen,

```
Pre-hematocrit:  40.00%
Est. Post-hematocrit:  27.60%
Age: 6 months +
W= 175 HT=  60 BSA=   1.76 QS=   4.05

Control ACT=  110 sec
Suggest 1st HEP dose 238 mg (3 mg/kg)
Actual 1st HEP: 240 mg

Est.2nd HEP for ACT=480 sec= 130 mg
For LIN. REGRESSION, actual 2nd HEP was
:  130 mg
Slope=  1.04
Corr Coeff= 0.999

ADDITIONAL DOSES:
(Current ACT is  350sec)
Heparin dose is  124 mg
Actual Heparin dose given:  150 mg
Total amount of Heparin given:   520 mg

(Current ACT is  450sec)
Protamine dose is  422 mg
Actual Protamine dose given:  450 mg
Total amount of Protamine given:   450
mg

Final ACT:   110 sec
Total Heparin given:   520 mg
Total Protamine given:   450 mg
```

Figure 3: A Computer printout illustrating the entered and calculated information contained within the program. The patients' hematocrit is printed followed by his age group, weight (W), height (H), body surface area (BSA), and estimated blood flow (QS). The control activated clotting time (ACT) is then given with a suggested heparin dose (3 mg/kg). The program then gives a recommended dosage in order to reach an ACT of 480 seconds. *Additional doses* are determined using the Bull protocol. Here the ACT is shown before every dose.

not the printer. Additionally, all data printed is dually displayed on the screen. Hence, the screen is primarily used for quick visual reference during program operation, whereas the data slip can be viewed at a later point in time.

The program (Figure 2) takes up 3,324 bytes of memory and is programmed in a form of BASIC, that is compatible with the ROM of the Sharp PC-2500. Other BASIC programs might need altering in accordance to the ROM of their system; such alterations could possibly change the total number of kilobytes used.

The calculations done by the program assist the perfusionist by eliminating the need to work out tedious and lengthy computations, thus providing him with time to devote elsewhere. Storage of data on paper negates the need to record items by hand, thus increasing efficiency and facilitating later review of data (Figure 3). The updated computer program has been utilized in more than 250 patients without problem.

Discussion

The present communication describes the evolution of a computerized technique to aid perfusionists and surgeons in doing rapid calculations relevant to cardiopulmonary bypass. Microcomputers are a rapidly developing part of operating room technology. Computing power becomes increasingly economical as component microcircuits are mass produced and developmental costs are absorbed. The capabilities of the microcomputer described here represent a dramatic advance in almost every respect over the programmable calculator described in our previous paper. Furthermore, the PC-2500 is less expensive than the calculator we used previously. It is likely that computing equipment will be increasingly relied upon during cardiopulmonary bypass and other forms of surgery to do complex calculations.

In this instance, our new equipment has eased the process of rapidly introducing changes into the program. The options that have been made available due to the increased memory and flexibility of microcomputers in comparison to calculators are many. Maintenance of the equipment also appears simplified, but the durability of the computer, especially the printer, remains to be proven.

Computer manufacturers appear reluctant to become involved in the custom manufacturing of devices for bypass computations because of low profit margins, wide individual variation in protocols for heart surgery, and potential product liability. For these reasons, we have utilized standard equipment to fit our needs; the flexibility of programmable microcomputers makes this possible.

Our previous experience with computing equipment has influenced our requirements for computing equipment to be used in the operating room. In the event of loss of power, battery power becomes a necessity; it is also helpful in eliminating clutter.

In our prior system, the process of reprogramming a calculator before every case with a magnetic card proved a tiresome chore that was subject to error, wear, and mechanical malfunction. Permanent storage of the program within the memory of the computer, maintained by low level battery power when the computer is turned off, eliminates this problem. A fully charged battery will function for approximately 15 hours (slightly less with extensive printer usage).[5] The battery is routinely recharged overnight, and we maintain three computers in the event of two simultaneous cases.

A permanent recording of the program on cassette tape is also an important element of the system because it allows the program to be restored quickly in the event of damage, which eliminates the need to retype and debug a new program. The audio tapes also simplify the process of exchanging programs with users in other institutions.

We wish to emphasize that any computer programmed for intraoperative use should be carefully debugged. The program should be fully tested before use with a variety of data to assure proper function.

The program must also be tested to be certain that it is immune to the personal idiosyncrasies of perfusionists and any others who are not intuitively computer-oriented. In addition, the users should be thoroughly trained in the operation of both the computer and the program. Operators must also be alert to the appearance of erroneous computed data and should be familiar with the procedures at hand in order to prevent mishaps from occurring.

We conclude that microcomputers are becoming an increasingly valuable adjunct to the conduct of cardiopulmonary bypass. The availability of increasing computer power intraoperatively will also permit the use of other computation-intensive procedures yet to be defined.

References

1. Rose, E.A., Haubert, S.M., Spotnitz, H.M.: Programmable Calculators and Cardiopulmonary Bypass. *J. Extra-Corpor. Technol.*, 1979; 11:5.
2. Du Bois, D., and Du Bois, E.F.: Clinical Calorimetry Tenth Paper: A Formula to Estimate the Approximate Surface Area if Height and Weight be Known. *Arch Int Med,* 1916, 16:863.
3. Bull, B.S., Huse, W.M., Brauer, F.S., and Korpman, R.A.: Heparin Therapy During Extracorporeal Circulation II. The Use of a Dose Response Curve to Individualize Heparin and Protamine Dosage, *J Thorac Cardiovasc Surg*, 1975; 69:685.
4. Bull, B.S., Korpman, R.A., Huse, W.M., and Briggs, B.D.: Heparin Therapy During Extracorporeal Circulation I. Problems Inherent in Existing Heparin Protocols. *J Thorac Cardiovasc Surg*, 1975; 69:674.
5. *Japan: Using the Print/Cassette Interface*. In Japan (ed): Sharp portable computer operation manual, Sharp Corporation 68-89, 1985.